Mathematical Model for 2D Point-footed, Midleg-Mass (Hipped) Walker

Modeling by Robert D Gregg
rdgregg@eecs.berkeley.edu
Center for Hybrid and Embedded Software Systems
Electrical Engineering and Computer Sciences
University of California, Berkeley

```
<< H:\Screws.m
<< H:\RobotLinks.m
```

Variables

```
x4 --> stance yaw angle
x5 --> stance roll angle
x6 --> stance pitch angle
x7 --> swing pitch angle
```

Constants

```
w --> hip width
l --> leg length
Mp --> pelvis / hip mass
M --> midleg mass
gamma --> slope angle
```

2D Walker Open Chain Kinematics

Foot/Leg links

```
q = {{x1[t]}, {x2[t]}, {x3[t]}, {x6[t]}, {x7[t]}};
qdot = D[q, t];
```

```
ax1 = {1, 0, 0, 0, 0, 0};
ax2 = {0, 1, 0, 0, 0, 0};
ax3 = {0, 0, 1, 0, 0, 0};
ax4 = {0, 0, 0, 0, 0, 1};
ax5 = {0, 0, 0, 0, 1, 0};
ax6 = {0, 0, 0, 1, 0, 0};
ax7 = Flatten[Append[Cross[{0, 0, l}, {1, 0, 0}], {1, 0, 0}]];

gst01 = {{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, l / 2}, {0, 0, 0, 1}};
gst02 = {{1, 0, 0, w / 2}, {0, 1, 0, 0}, {0, 0, 1, l}, {0, 0, 0, 1}};
gst03 = {{1, 0, 0, w}, {0, 1, 0, 0}, {0, 0, 1, l / 2}, {0, 0, 0, 1}};

J1 = BodyJacobian[{ax1, x1[t]}, {ax2, x2[t]}, {ax3, x3[t]},
    {ax6, x6[t]}, {{0, 0, 0, 0, 0, 0}, x7[t]}, gst01] // FullSimplify;
J2 = BodyJacobian[{ax1, x1[t]}, {ax2, x2[t]}, {ax3, x3[t]}, {ax6, x6[t]},
    {{0, 0, 0, 0, 0, 0}, x7[t]}, gst02] // FullSimplify;
J3 = BodyJacobian[{ax1, x1[t]}, {ax2, x2[t]}, {ax3, x3[t]},
    {ax6, x6[t]}, {ax7, x7[t]}, gst03] // FullSimplify;
MatrixForm[J1]
MatrixForm[J2]
MatrixForm[J3]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & \text{Cos}[x6[t]] & \text{Sin}[x6[t]] & -\frac{l}{2} & 0 \\ 0 & -\text{Sin}[x6[t]] & \text{Cos}[x6[t]] & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & \text{Cos}[x6[t]] & \text{Sin}[x6[t]] & -l & 0 \\ 0 & -\text{Sin}[x6[t]] & \text{Cos}[x6[t]] & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & \text{Cos}[x6[t]+x7[t]] & \text{Sin}[x6[t]+x7[t]] & \frac{1}{2}(l-2\,l\,\text{Cos}[x7[t]]) & \frac{l}{2} \\ 0 & -\text{Sin}[x6[t]+x7[t]] & \text{Cos}[x6[t]+x7[t]] & l\,\text{Sin}[x7[t]] & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

```
genM1 = M * {{1, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0},
    {0, 0, 1, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}};
genM2 = Mp * {{1, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0}, {0, 0, 1, 0, 0, 0},
    {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}};
genM3 = M * {{1, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0}, {0, 0, 1, 0, 0, 0},
    {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}};
```

```
MatrixForm[Mmatrix = Transpose[J1].genM1.J1 +
    Transpose[J2].genM2.J2 + Transpose[J3].genM3.J3 // FullSimplify]
```

$$\begin{pmatrix} 2\,M + Mp & 0 & 0 \\ 0 & 2\,M + Mp & 0 \\ 0 & 0 & 2\,M + Mp \\ 0 & \frac{1}{2}\,l\,(-(3\,M + 2\,Mp)\,Cos[x6[t]] + M\,Cos[x6[t] + x7[t]]) & \frac{1}{2}\,l\,(-(3\,M + 2\,Mp)\,Sin[x6[t]] + M \\ 0 & \frac{1}{2}\,l\,M\,Cos[x6[t] + x7[t]] & \frac{1}{2}\,l\,M\,Sin[x6[t] + x7[t]] \end{pmatrix}$$

Without a hip (the same thing because this is a planar walker):

```
MatrixForm[Mmatrix /. w → 0 // FullSimplify];
```

```
KE = First[First[1 / 2 Transpose[qdot].Mmatrix.qdot]] // FullSimplify
```

$$\frac{1}{8}\,(4\,(2\,M + Mp)\,x1'[t]^2 + 4\,(2\,M + Mp)\,x2'[t]^2 + 8\,M\,x3'[t]^2 + 4\,Mp\,x3'[t]^2 - 12\,l\,M\,Sin[x6[t]]\,x3'[t]\,x6'[t] - 8\,l\,Mp\,Sin[x6[t]]\,x3'[t]\,x6'[t] + 6\,l^2\,M\,x6'[t]^2 + 4\,l^2\,Mp\,x6'[t]^2 + 2\,l^2\,M\,x6'[t]\,x7'[t] + l^2\,M\,x7'[t]^2 + 4\,l\,M\,Cos[x6[t]]\,Sin[x7[t]]\,x3'[t]\,(x6'[t] + x7'[t]) + 4\,l\,M\,Cos[x7[t]]\,(Sin[x6[t]]\,x3'[t] - l\,x6'[t])\,(x6'[t] + x7'[t]) + 4\,l\,x2'[t]\,((-(3\,M + 2\,Mp)\,Cos[x6[t]] + M\,Cos[x6[t] + x7[t]])\,x6'[t] + M\,Cos[x6[t] + x7[t]]\,x7'[t]))$$

```
KE /. {x1[t] → 0, x2[t] → 0, x3[t] → 0, x1'[t] → 0, x2'[t] → 0, x3'[t] → 0} // FullSimplify
```

$$\frac{1}{8}\,l^2\,((6\,M + 4\,Mp - 4\,M\,Cos[x7[t]])\,x6'[t]^2 + 2\,M\,(1 - 2\,Cos[x7[t]])\,x6'[t]\,x7'[t] + M\,x7'[t]^2)$$

Potential Term

```
MatrixForm[e1 = TwistExp[ax1, x1[t]]]
MatrixForm[e2 = TwistExp[ax2, x2[t]]]
MatrixForm[e3 = TwistExp[ax3, x3[t]]]
MatrixForm[e4 = TwistExp[ax4, x4[t]]];
MatrixForm[e5 = TwistExp[ax5, x5[t]]];
MatrixForm[e6 = TwistExp[ax6, x6[t]]]
MatrixForm[e7 = TwistExp[ax7, x7[t]]]
```

$$\begin{pmatrix} 1 & 0 & 0 & x1[t] \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & x2[t] \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & x3[t] \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[x6[t]] & -\mathrm{Sin}[x6[t]] & 0 \\ 0 & \mathrm{Sin}[x6[t]] & \mathrm{Cos}[x6[t]] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[x7[t]] & -\mathrm{Sin}[x7[t]] & l\,\mathrm{Sin}[x7[t]] \\ 0 & \mathrm{Sin}[x7[t]] & \mathrm{Cos}[x7[t]] & l\,(1-\mathrm{Cos}[x7[t]]) \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
pS = {{0}, {0}, {l / 2}, {1}};
pH = {{w / 2}, {0}, {l}, {1}};
pNS = {{w}, {0}, {l / 2}, {1}};
```

```
MatrixForm[pSp = e1.e2.e3.e6.pS // FullSimplify]
MatrixForm[pHp = e1.e2.e3.e6.pH // FullSimplify]
MatrixForm[pNSp = e1.e2.e3.e6.e7.pNS // FullSimplify]
```

$$\begin{pmatrix} x1[t] \\ -\frac{1}{2}\, l\, \mathrm{Sin}[x6[t]] + x2[t] \\ \frac{1}{2}\, l\, \mathrm{Cos}[x6[t]] + x3[t] \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{w}{2} + x1[t] \\ -l\, \mathrm{Sin}[x6[t]] + x2[t] \\ l\, \mathrm{Cos}[x6[t]] + x3[t] \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} w + x1[t] \\ \frac{1}{2}\, l\, (-2\, \mathrm{Sin}[x6[t]] + \mathrm{Sin}[x6[t] + x7[t]]) + x2[t] \\ l\, \mathrm{Cos}[x6[t]] - \frac{1}{2}\, l\, \mathrm{Cos}[x6[t] + x7[t]] + x3[t] \\ 1 \end{pmatrix}$$

```
PE = First[M * g * pSp[[3]] + Mp * g * pHp[[3]] + M * g * pNSp[[3]]] // FullSimplify
```

$$\frac{1}{2}\, g\, (l\, (3\, M + 2\, Mp)\, \mathrm{Cos}[x6[t]] - l\, M\, \mathrm{Cos}[x6[t] + x7[t]] + 2\, (2\, M + Mp)\, x3[t])$$

```
% /. w → 0
```

$$\frac{1}{2}\, g\, (l\, (3\, M + 2\, Mp)\, \mathrm{Cos}[x6[t]] - l\, M\, \mathrm{Cos}[x6[t] + x7[t]] + 2\, (2\, M + Mp)\, x3[t])$$

3D Walker Dynamics

```
Lagrangian = KE - PE /.
    {x1[t] → 0, x2[t] → 0, x3[t] → 0, x1'[t] → 0, x2'[t] → 0, x3'[t] → 0} // FullSimplify;
```

```
eq1 = D[D[Lagrangian, x6'[t]], t] - D[Lagrangian, x6[t]] // FullSimplify
eq2 = D[D[Lagrangian, x7'[t]], t] - D[Lagrangian, x7[t]] // FullSimplify
```

$$\frac{1}{4}\, l\, (-2\, g\, (3\, M + 2\, Mp)\, \mathrm{Sin}[x6[t]] + 2\, g\, M\, \mathrm{Sin}[x6[t] + x7[t]] + l\, (2\, M\, \mathrm{Sin}[x7[t]]\, x7'[t]\, (2\, x6'[t] + x7'[t]) + 6\, M\, x6''[t] + 4\, Mp\, x6''[t] + M\, x7''[t] - 2\, M\, \mathrm{Cos}[x7[t]]\, (2\, x6''[t] + x7''[t])))$$

$$\frac{1}{4}\, l\, M\, (2\, g\, \mathrm{Sin}[x6[t] + x7[t]] + l\, (-2\, \mathrm{Sin}[x7[t]]\, x6'[t]^2 + (1 - 2\, \mathrm{Cos}[x7[t]])\, x6''[t] + x7''[t]))$$

```
Solve[{eq1 == 0 , eq2 == 0}, {x6''[t], x7''[t]}] // FullSimplify
```

```
{{x6''[t] → (2 (-2 g (M + Mp) Sin[x6[t]] + g M Sin[x6[t] + 2 x7[t]] +
        l M (-Sin[2 x7[t]] x6'[t]^2 + Sin[x7[t]] (x6'[t] + x7'[t])^2))) /
    (l (-3 M - 4 Mp + 2 M Cos[2 x7[t]])), x7''[t] → 1/(l (-3 M - 4 Mp + 2 M Cos[2 x7[t]]))
    (-2 g (-2 (M + Mp) + M (Cos[x7[t]] + Cos[2 x7[t]])) Sin[x6[t]] +
      2 g Cos[x6[t]] (5 M + 4 Mp - 2 M Cos[x7[t]]) Sin[x7[t]] +
      4 l (-3 M - 2 Mp + 2 M Cos[x7[t]]) Sin[x7[t]] x6'[t]^2 + 4 l M (-Sin[x7[t]] + Sin[2 x7[t]])
       x6'[t] x7'[t] + 2 l M (-Sin[x7[t]] + Sin[2 x7[t]]) x7'[t]^2)}}
```

■ Without a hip (the same thing because this is a planar walker):

```
Solve[{eq1 == 0 /. w → 0, eq2 == 0 /. w → 0}, {x6''[t], x7''[t]}] // FullSimplify;
```

Impact Equations

```
pNSFoot = {{w}, {0}, {0}, {1}};
MatrixForm[pNSFootp = e1.e2.e3.e6.e7.pNSFoot // FullSimplify]
```

$$\begin{pmatrix} w + x1[t] \\ l\,(-\mathrm{Sin}[x6[t]] + \mathrm{Sin}[x6[t] + x7[t]]) + x2[t] \\ l\,(\mathrm{Cos}[x6[t]] - \mathrm{Cos}[x6[t] + x7[t]]) + x3[t] \\ 1 \end{pmatrix}$$

```
Pos1 = First[pNSFootp[[1]]];
Pos2 = First[pNSFootp[[2]]];
Pos3 = First[pNSFootp[[3]]];
MatrixForm[F = FullSimplify[
   {{D[Pos1, x1[t]], D[Pos1, x2[t]], D[Pos1, x3[t]], D[Pos1, x6[t]], D[Pos1, x7[t]]},
    {D[Pos2, x1[t]], D[Pos2, x2[t]], D[Pos2, x3[t]], D[Pos2, x6[t]], D[Pos2, x7[t]]},
    {D[Pos3, x1[t]], D[Pos3, x2[t]], D[Pos3, x3[t]], D[Pos3, x6[t]], D[Pos3, x7[t]]}}]]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & l\,(-\mathrm{Cos}[x6[t]] + \mathrm{Cos}[x6[t] + x7[t]]) & l\,\mathrm{Cos}[x6[t] + x7[t]] \\ 0 & 0 & 1 & l\,(-\mathrm{Sin}[x6[t]] + \mathrm{Sin}[x6[t] + x7[t]]) & l\,\mathrm{Sin}[x6[t] + x7[t]] \end{pmatrix}$$

```
MatrixForm[nFT = Join[-Transpose[F], {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}]]
```

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \\ 0 & -l\,(-\mathrm{Cos}[x6[t]] + \mathrm{Cos}[x6[t] + x7[t]]) & -l\,(-\mathrm{Sin}[x6[t]] + \mathrm{Sin}[x6[t] + x7[t]]) \\ 0 & -l\,\mathrm{Cos}[x6[t] + x7[t]] & -l\,\mathrm{Sin}[x6[t] + x7[t]] \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
MatrixForm[FullMatrix = Transpose[Join[Transpose[Join[Mmatrix, F]], Transpose[nFT]]]]
```

$$\begin{pmatrix} 2M+Mp & 0 & 0 \\ 0 & 2M+Mp & 0 \\ 0 & 0 & 2M+Mp \\ 0 & \frac{1}{2} l\,(-(3M+2Mp)\,Cos[x6[t]] + M\,Cos[x6[t]+x7[t]]) & \frac{1}{2} l\,(-(3M+2Mp)\,Sin[x6[t]] + M \\ 0 & \frac{1}{2} l\,M\,Cos[x6[t]+x7[t]] & \frac{1}{2} l\,M\,Sin[x6[t]+x7[t]] \\ 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

■ Without a hip (the same thing because this is a planar walker):

```
MatrixForm[FullMatrix /. w → 0 // FullSimplify];

MatrixForm[DMatrix = Join[Mmatrix.{0, 0, 0, x6'[t], x7'[t]}, {0, 0, 0}]]
```

$$\begin{pmatrix} 0 \\ \frac{1}{2} l\,(-(3M+2Mp)\,Cos[x6[t]] + M\,Cos[x6[t]+x7[t]])\,x6'[t] + \frac{1}{2} l\,M\,Cos[x6[t]+x7[t]]\,x7'[t] \\ \frac{1}{2} l\,(-(3M+2Mp)\,Sin[x6[t]] + M\,Sin[x6[t]+x7[t]])\,x6'[t] + \frac{1}{2} l\,M\,Sin[x6[t]+x7[t]]\,x7'[t] \\ \frac{1}{2} l^2\,(3M+2Mp-2M\,Cos[x7[t]])\,x6'[t] + \frac{1}{4} l\,M\,(l-2\,l\,Cos[x7[t]])\,x7'[t] \\ \frac{1}{4} l\,M\,(l-2\,l\,Cos[x7[t]])\,x6'[t] + \frac{1}{4} l^2\,M\,x7'[t] \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

```
(% FullMatrix = FullMatrix /. w → 0 // FullSimplify;
 DMatrix = DMatrix /. w → 0 // FullSimplify; %)

K = Inverse[FullMatrix].DMatrix;
```

First, we have the angular positions for 6,7:

```
{x6impact = x6[t] + x7[t], x7impact = -x7[t]};
```

We are interested in the angular velocities for 6,7:

```
x6dotimpact = FullSimplify[K[[4]] + K[[5]]]
x7dotimpact = -FullSimplify[K[[5]]]
```

$$\frac{(M - 2\,(M + 2\,Mp)\,Cos[x7[t]])\,x6'[t] + M\,x7'[t]}{-3\,M - 4\,Mp + 2\,M\,Cos[2\,x7[t]]}$$

$$-\frac{-4\,(M + Mp)\,(Cos[x7[t]] - Cos[2\,x7[t]])\,x6'[t] + M\,(1 - 2\,Cos[x7[t]])\,x7'[t]}{-3\,M - 4\,Mp + 2\,M\,Cos[2\,x7[t]]}$$

■ Collision Guard

Note: gamma is the slope angle

```
height = First[pNSFootp[[3]] + Tan[gamma] * pNSFootp[[2]] /. {x2[t] → 0, x3[t] → 0}]

l (Cos[x6[t]] - Cos[x6[t] + x7[t]]) + l (-Sin[x6[t]] + Sin[x6[t] + x7[t]]) Tan[gamma]

Avect = {{D[height, x6[t]], D[height, x7[t]]}};
holonomicTraj = First[First[Avect.{{x6'[t]}, {x7'[t]}}]] // FullSimplify

l Sec[gamma]
 ((-Sin[gamma + x6[t]] + Sin[gamma + x6[t] + x7[t]]) x6′[t] + Sin[gamma + x6[t] + x7[t]] x7′[t])
```

The guard is the zero-level set of the height function:

```
height == 0

l (Cos[x6[t]] - Cos[x6[t] + x7[t]]) + l (-Sin[x6[t]] + Sin[x6[t] + x7[t]]) Tan[gamma] == 0
```

and the negative region of the holonomic constraint's trajectory:

```
holonomicTraj < 0

l Sec[gamma] ((-Sin[gamma + x6[t]] + Sin[gamma + x6[t] + x7[t]]) x6′[t] +
    Sin[gamma + x6[t] + x7[t]] x7′[t]) < 0
```